CARS, INTERCHANGES, TRAFFIC COUNTERS, AND SOME PRETTY DARNED GOOD KNOT INVARIANTS

DROR BAR-NATAN, UNIVERSITY OF TORONTO

ABSTRACT. A condensed summary of a talk I gave in Nara on August 13, 2023: Reporting on joint work with Roland van der Veen, I'll tell you some stories about ρ_1, an easy to define, strong, fast to compute, homomorphic, and well-connected knot invariant. ρ_1 was first studied by Rozansky and Overbay [Ro1, Ro2, Ro3, Ov] and Ohtsuki [Oh2], it has far-reaching generalizations, it is dominated by the coloured Jones polynomial, and I wish I understood it.

My talk's title and abstract were the same as the title and abstract of this summary. The talk used slides, and in this summary, they are shown on the right.

Those slides were all excerpts from a handout, which is attached at the end of this document. It is where the true content lies! It is also available on the web site of this talk, which is displayed on the next slide.

As an aside, I really believe in this way of giving talks, with slides and a handout. Slides are to save time and to allow for more elaborate figures. But slide talks without a handout are awful! Content disappears before it's been digested. A handout with identical content to the slides solves the problem – you can always look back to recall (and ahead, to decide how hard you want to fight sleep). But then the best way to make sure that the handout and the slides are fully synced is to have the slides simply be zoomed-in parts of the handout, and that's precisely what I do.

But it's a waste of so much paper, I hear you say. Yes, I say, but it's completely trivial relative to our travel to hear each other talk. Save where it matters. Where it's useful, spend.

Okay, it's all online, at ωεβ:=http://drorbn.net/na23. There's also a paper, at ωεβ/APAI.



Thanks, NSERC and Arthur Chu!



Date: August 14, 2023.

We seek strong, fast, and homomorphic invariants. Strong and fast are clear enough. Especially, we care for fast because of the likes of the GST48 knot [GST] and the Piccirillo knot [Pi]. Polynomial time is best!



We then explained "homomorphic". It means, "extends to tangles and is well behaved under tangle gluing and strand doubling".



We care for "homomorphic" because using tangles and tangle operations we can define interesting classes of knots, and thus invariants that are homomorphic with respect to these operations may be able to tell us something about these classes. See ωεβ/AKT.



But enough with philosophy! I learned from Vaughan Jones that theories change with time, yet formulas stay. So let's start with formulas!



To compute our knot invariant ρ_1, we cut it to a long knot and place it in the plane so that at all vertices, all edges are "flowing up". We then label each edge with serial number and with its rotation number φ_k.



We make a $(2n+1)\times(2n+1)$ matrix A by starting with the identity matrix and adding a 2×2 block for each crossing, as shown on the right. We let $G = (g_{\alpha\beta})$ be the inverse of A.



If we start from the trefoil knot diagram displayed before, the resulting A is shown on the right.



And now the corresponding G, the "Green Function", is shown.



We noted that $\det(A)$ is (up to a normalization) the good old Alexander polynomial. If you are a classical topologist, you should yawn and perhaps fall asleep right now, for so far everything is very old material.



The 2×2 matrices are the Burau matrices. The matrix A is a presentation matrix of the Alexander module, derived by applying Fox calculus to the Wirtinger presentation. Even G is not a great surprise; it is related to the "Blanchfield Pairing". All of these people are old timers, so much so that their pictures are in black and white.

All the news is in just one slide, the one on the right! We defined $R_1(c)$ and ρ_1, explained why ρ_1 is easy to compute (as easy as the Alexander polynomial), and asserted that it is invariant (to be proven below). If you are a classical topologist, these formulas should come as a complete surprise to you.

Formulas, continued. Finally, set

$$R_1(c) := s\left(g_{ji}\left(g_{j+1,j} + g_{j,j+1} - g_{ij}\right) - g_{ii}\left(g_{j,j+1} - 1\right) - 1/2\right)$$

$$\rho_1 := \Delta^2\left(\sum_c R_1(c) - \sum_k \varphi_k\,(g_{kk} - 1/2)\right).$$

In our example $\rho_1 = -T^2 + 2T - 2 + 2T^{-1} - T^{-2}$.

Theorem. ρ_1 is a knot invariant. Proof: later.

Classical Topologists: Whiskey Tango Foxtrot?

These days I take what I learned from Vaughan Jones a step further. I care for programs even more than I care for formulas.

Jones:
Formulas stay;
interpretations change with time.

We load some libraries that play a mild role: just tables of knots, and some older invariants for comparison, and a program to compute rotation numbers (something we could have done by hand).

Preliminaries

This is Rho.nb of http://drorbn.net/oa22/ap.

```
Once[<< KnotTheory`; << Rot.m];

Loading KnotTheory` version
  of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.

Loading Rot.m from http://drorbn.net/la22/ap
  to compute rotation numbers.
```

Next is the main part of the program. It is almost one-to-one the same as the formulas for ρ_1, and if there's ever a disagreement, the program is to be trusted better because it's been tested extensively. Note that the program outputs the ordered pair $Z = (\Delta, \rho_1)$, because Δ is computed anyway within the computation of ρ_1, and we consider it as a part of ρ_1.

The Program

```
R₁[s_, i_, j_] :=
  s (g_{ji} (g_{j⁺,j} + g_{j,j⁺} - g_{ij}) - g_{ii} (g_{j,j⁺} - 1) - 1/2);
Z[K_] := Module[{Cs, φ, n, A, s, i, j, k, Δ, G, ρ1},
   {Cs, φ} = Rot[K]; n = Length[Cs];
   A = IdentityMatrix[2 n + 1];
   Cases[Cs, {s_, i_, j_} ⧴
     (A[[{i, j}, {i + 1, j + 1}]] += ( -T^s  T^s - 1 ;  0  -1 ))];
   Δ = T^((-Total[φ] - Total[Cs[[All,1]]])/2) Det[A];
   G = Inverse[A];
   ρ1 = Sum_{k=1}^{n} R₁ @@ Cs[[k]] - Sum_{k=1}^{2 n} φ[[k]] (g_{kk} - 1/2);
   Factor@
    {Δ, Δ^2 ρ1 /. α_⁺ ⧴ α + 1 /. g_{α_,β_} ⧴ G[[α, β]]}];
```

We run the program on all knots with up to 6 crossings.

The First Few Knots

```
TableForm[Table[Join[{K[[1]]_K[[2]]}, Z[K]],
  {K, AllKnots[{3, 6}]}], TableAlignments → Center]
```

3_1	$\frac{1-T+T^2}{T}$	$\frac{(-1+T)^2(1+T^2)}{T^2}$
4_1	$-\frac{1-3T+T^2}{T}$	0
5_1	$\frac{1-T+T^2-T^3+T^4}{T^2}$	$\frac{(-1+T)^2(1+T^2)(2+T^2+2T^4)}{T^4}$
5_2	$\frac{2-3T+2T^2}{T}$	$\frac{(-1+T)^2(5-4T+5T^2)}{T^2}$
6_1	$-\frac{(-2+T)(-1+2T)}{T}$	$\frac{(-1+T)^2(1-4T+T^2)}{T^2}$
6_2	$-\frac{1-3T+3T^2-3T^3+T^4}{T^2}$	$\frac{(-1+T)^2(1-4T+4T^2-4T^3+4T^4-4T^5+T^6)}{T^4}$
6_3	$\frac{1-3T+5T^2-3T^3+T^4}{T^2}$	0

The program is fast! Here is the GST48 knot once again,...

Fast!



and it takes only about 170 seconds to compute its ρ_1.

```
Timing@
 Z[GST48 = EPD[X14,1, X̄2,29, X3,40, X43,4, X̄26,5, X6,95,
    X96,7, X13,8, X̄9,28, X10,41, X42,11, X̄27,12, X30,15,
    X̄16,61, X̄17,72, X̄18,83, X19,34, X̄89,20, X̄21,92,
    X̄79,22, X̄68,23, X̄57,24, X̄25,56, X62,31, X73,32,
    X84,33, X̄50,35, X36,81, X37,70, X38,59, X̄39,54, X44,55,
    X58,45, X69,46, X80,47, X48,91, X90,49, X51,82, X52,71,
    X53,60, X̄63,74, X̄64,85, X̄76,65, X̄87,66, X̄67,94,
    X̄75,86, X̄88,77, X̄78,93]]
```

$$\Big\{170.313, \Big\{-\frac{1}{T^8}(-1+2T-T^2-T^3+2T^4-T^5+T^8)(-1+T^3-2T^4+T^5+T^6-2T^7+T^8),\ \frac{1}{T^{16}}(-1+T)^2(5-18T+33T^2-32T^3+2T^4+42T^5-62T^6-8T^7+166T^8-242T^9+108T^{10}+132T^{11}-226T^{12}+148T^{13}-11T^{14}-36T^{15}-11T^{16}+148T^{17}-226T^{18}+132T^{19}+108T^{20}-242T^{21}+166T^{22}-8T^{23}-62T^{24}+42T^{25}+2T^{26}-32T^{27}+33T^{28}-18T^{29}+5T^{30})\Big\}\Big\}$$

$Z = (\Delta, \rho_1)$ is strong! It seems that it is stronger than HOMFLY-PT and Khovanov homology taken together.

Strong!

```
{NumberOfKnots[{3, 12}],
 Length@
  Union@Table[Z[K], {K, AllKnots[{3, 12}]}],
 Length@
  Union@Table[{HOMFLYPT[K], Kh[K]},
    {K, AllKnots[{3, 12}]}]}
```

{2977, 2882, 2785}

So the pair (Δ, ρ_1) attains 2,882 distinct values on the 2,977 prime knots with up to 12 crossings (a deficit of 95), whereas the pair (HOMFLYPT, Khovanov Homology) attains only 2,785 distinct values on the same knots (a deficit of 192).

On to interpretations, we discussed the traffic rules for cars on a knot diagram. All car crashes we discuss are gentle and no harm is ever caused to the occupants of our cars.



A relevant scene with Lightning McQueen, enacted by Roland's kids.



We claim that the matrix G of before is the traffic matrix for a knot diagram. Yet first we illustrate the traffic matrix using a very simple knot diagram (a single kink), and some simple-minded geometric summation.



We then used "g-rules" to prove the claim. These are rules that tell us how to move the traffic injection sites and the traffic counting sites, and they will also be useful below, within the actual proof of invariance.



On to the invariance under the hardest of the Reidemeister moves, Reid3. We first establish that traffic away from the Reid3 site is not affected by the move. This is essentially the invariance of the Burau representation.



It follows that we only need to understand the contribution of the $R_1(c)$ terms from the crossings within the Reid3 area.



We could have done it by hand, but we are lazy and we have good computer skills. So we type in the g-rules, the three R_1 contributions for the left hand side of Reid3 and the three R_1 contributions for the right hand side. We then apply the g-rules to move the traffic injection sites and the traffic counting sites to outside of the Reid3-move area, to where they are unchanged by the move. Comparing `lhs` with `rhs`, the computer says `True`, which means that ρ_1 is invariant under Reid3.



As a second example we verify invariance under Reid1. Most of the work had already been done, because we computed already the "traffic matrix" of a kink. What remains is a little calculation (without forgetting the rotation-number correction!). We do that calculation on the right, using a hybrid of computer and human power (very little of each). A few further



moves need to be shown — they are discussed at ωεβ/APAI. This concludes the invariance proof for ρ_1.

This slide ought to be shown bigger. Wearing my topology hat, I genuinely, honestly, don't know what's going on.



Unfortunately, at this point we had to rush towards the end, and be brief. Wearing my quantum algebra hat, the first thing to note is that there is a whiff of a Heisenberg relation in car traffic — a difference of 1 between the traffic counting before and after the place where traffic is injected, and that may remind us of the Heisenberg commutation relation, $[p, x] = 1$.



I have gone through the remaining few slides way too quickly. Here I will let them speak for themselves. The main things to learn from these reproduced slides are the references cited in them, and the comments in red.

Where did it come from? Consider $\mathfrak{g}_\epsilon := sl^\epsilon_{2+} := L\langle y, b, a, x\rangle$ with relations

$$[b, x] = \epsilon x, \quad [b, y] = -\epsilon y, \quad [b, a] = 0,$$
$$[a, x] = x, \quad [a, y] = -y, \quad [x, y] = b + \epsilon a.$$

At invertible ϵ, it is isomorphic to sl_2 plus a central factor, and it can be quantized à la Drinfel'd [Dr] much like sl_2 to get an algebra $QU = A\langle y, b, a, x\rangle$ subject to (with $q = e^{\hbar\epsilon}$):

$$[b, a] = 0, \quad [b, x] = \epsilon x, \quad [b, y] = -\epsilon y,$$
$$[a, x] = x, \quad [a, y] = -y,$$
$$xy - qyx = \frac{1 - e^{-\hbar(b+\epsilon a)}}{\hbar}.$$

Now QU has an R-matrix solving Yang-Baxter (meaning Reid3),

$$R = \sum_{m,n\geq 0} \frac{y^n b^m \otimes (\hbar a)^m (\hbar x)^n}{m![n]_q!},$$

($[n]_q!$ is a "quantum factorial")

and so it has an associated "universal quantum invariant" à la Lawrence and Ohtsuki [La, Oh1], $Z_\epsilon(K) \in QU$.

Now $QU \cong \mathcal{U}(\mathfrak{g}_\epsilon)$ (only as algebras!) and $\mathcal{U}(\mathfrak{g}_\epsilon)$ represents into $\mathbb{H}$ via

$$y \to -tp - \epsilon \cdot xp^2, \quad b \to t + \epsilon \cdot xp,$$
$$a \to xp, \quad x \to x,$$

(abstractly, $\mathfrak{g}_\epsilon$ acts on its Verma module

$$\mathcal{U}(\mathfrak{g}_\epsilon)/(\mathcal{U}(\mathfrak{g}_\epsilon)\langle y, a, b - \epsilon a - t\rangle) \cong \mathbb{Q}[x]$$

by differential operators, namely via $\mathbb{H}$), so R can be pushed to $\mathcal{R} \in \mathbb{H} \otimes \mathbb{H}$.

Everything still makes sense at $\epsilon = 0$ and can be expanded near $\epsilon = 0$ resulting with $\mathcal{R} = \mathcal{R}_0(1 + \epsilon\mathcal{R}_1 + \cdots)$, with $\mathcal{R}_0 = e^{t(xp\otimes 1 - x\otimes p)}$ and $\mathcal{R}_1$ a quartic polynomial in p and x. So p's and x's get created along K and need to be pushed around to a standard location ("normal ordering"). This is done using

$$(p \otimes 1)\mathcal{R}_0 = \mathcal{R}_0(T(p \otimes 1) + (1 - T)(1 \otimes p)),$$
$$(1 \otimes p)\mathcal{R}_0 = \mathcal{R}_0(1 \otimes p),$$

and when the dust settles, we get our formulas for ρ_1. But QU is a quasi-triangular Hopf algebra, and hence **ρ_1 is homomorphic**. Read more at [BV1, BV2] and hear more at ωεβ/SolvApp, ωεβ/Dogma, ωεβ/DoPeGDO, ωεβ/FDA, ωεβ/AQDW.


Schaveling

Also, we can (and know how to) look at higher powers of ϵ and we can (and more or less know how to) replace sl_2 by arbitrary semi-simple Lie algebra (e.g., [Sch]). So **ρ_1 is not alone!**

These constructions are very similar to Rozansky-Overbay [Ro1, Ro2, Ro3, Ov] and hence to the "loop expansion" of the Kontsevich integral and the coloured Jones polynomial [Oh2].

We re-iterated that an invariant as simple as ρ_1 must have a simple explanation, hopefully, within topology. Our current understanding of ρ_1 within quantum algebra is simply way too complicated.

If this all reads like **insanity** to you, it should (and you haven't seen half of it). Simple things should have simple explanations. Hence, **Homework.** Explain ρ_1 with no reference to quantum voodoo and find it a topology home (large enough to house generalizations!). Make explicit the homomorphic properties of ρ_1. Use them to do topology!

We also remind that in some sense, ρ_1 is a "friend" of the Alexander polynomial Δ, and that Δ is perhaps the most topologically-meaningful knot invariant. Like Δ, ρ_1 gives a genus bound. Does it also give a ribbon criteria like the Fox-Milnor condition for Δ?

P.S. As a friend of Δ, ρ_1 gives a genus bound, sometimes better than Δ's. How much further does this friendship extend?

At the end, we merely flashed our theorem regarding ρ_d, which generalizes ρ_1 when $d \geq 1$, and our implementation thereof. For $d \geq 2$, ρ_d is more complicated than ρ_1, yet it retains some things in common with ρ_1: Once more the key is the matrix $G = (g_{\alpha\beta})$. To compute ρ_1 we carry out a 1-fold summation over the features of the knot (crossings and rotations), of polynomials of degree ≤ 2 in the $g_{\alpha\beta}$'s. To compute ρ_d we carry out a d-fold summation over the features of the knot, of polynomials of degree $\leq 2d$ in the $g_{\alpha\beta}$'s. Multiple summations are of course more costly than single summations, yet the computation of ρ_2 remains of polynomial time and for small d it is completely practical.

References

[BV1] D. Bar-Natan and R. van der Veen, *A Polynomial Time Knot Polynomial,* Proc. Amer. Math. Soc. **147** (2019) 377–397, arXiv:1708.04853.

[BV2] D. Bar-Natan and R. van der Veen, *Perturbed Gaussian Generating Functions for Universal Knot Invariants,* arXiv:2109.02057.

[Dr] V. G. Drinfel'd, *Quantum Groups,* Proc. Int. Cong. Math., 798–820, Berkeley, 1986.

[GST] R. E. Gompf, M. Scharlemann, and A. Thompson, *Fibered Knots and Potential Counterexamples to the Property 2R and Slice-Ribbon Conjectures,* Geom. and Top. **14** (2010) 2305–2347, arXiv:1103.1601.

[Jo] V. F. R. Jones, *Hecke Algebra Representations of Braid Groups and Link Polynomials,* Annals Math., **126** (1987) 335-388.

[La] R. J. Lawrence, *Universal Link Invariants using Quantum Groups,* Proc. XVII Int. Conf. on Diff. Geom. Methods in Theor. Phys., Chester, England, August 1988. World Scientific (1989) 55–63.

[LTW] X-S. Lin, F. Tian, and Z. Wang, *Burau Representation and Random Walk on String Links,* Pac. J. Math., **182-2** (1998) 289–302, arXiv:q-alg/9605023.

[Oh1] T. Ohtsuki, *Quantum Invariants,* Series on Knots and Everything **29**, World Scientific 2002.

[Oh2] T. Ohtsuki, *On the 2–loop Polynomial of Knots,* Geom. Top. **11** (2007) 1357–1475.

[Ov] A. Overbay, *Perturbative Expansion of the Colored Jones Polynomial,* Ph.D. thesis, University of North Carolina, August 2013, ωεβ/Ov.

[Pi] L. Piccirillo, *The Conway knot is not slice,* Ann. of Math. (2) **191-2** (2020) 581–591, arXiv:1808.02923.

[Ro1] L. Rozansky, *A Contribution of the Trivial Flat Connection to the Jones Polynomial and Witten's Invariant of 3D Manifolds, I,* Comm. Math. Phys. **175-2** (1996) 275–296, arXiv:hep-th/9401061.

[Ro2] L. Rozansky, *The Universal R-Matrix, Burau Representation and the Melvin-Morton Expansion of the Colored Jones Polynomial,* Adv. Math. **134-1** (1998) 1–31, arXiv:q-alg/9604005.

[Ro3] L. Rozansky, *A Universal* $U(1)$*-RCC Invariant of Links and Rationality Conjecture,* arXiv: math/0201139.

[Sch] S. Schaveling, *Expansions of Quantum Group Invariants,* Ph.D. thesis, Universiteit Leiden, September 2020, ωεβ/Scha.

Email address: drorbn@math.toronto.edu
URL: http://www.math.toronto.edu/~drorbn

Cars, Interchanges, Traffic Counters, and some Pretty Darned Good Knot Invariants

More at ωεβ/APAI

Abstract. Reporting on joint work with Roland van der Veen, I'll tell you some stories about ρ_1, an easy to define, strong, fast to compute, homomorphic, and well-connected knot invariant. ρ_1 was first studied by Rozansky and Overbay [Ro1, Ro2, Ro3, Ov] and Ohtsuki [Oh2], it has far-reaching generalizations, it is dominated by the coloured Jones polynomial, and I wish I understood it. **Common misconception.** "Dominated" $\not\Rightarrow$ "lesser".

Rozansky Overbay Ohtsuki van der Veen

We seek strong, fast, homomorphic knot and tangle invariants.
Strong. Having a small "kernel".
Fast. Computable even for large knots (best: poly time).





Piccirillo

Gompf–Scharlemann–Thompson

Homomorphic. Extends to tangles and behaves under tangle operations; especially gluings and doublings:



Why care for "Homomorphic"? Theorem. A knot K is *ribbon* iff there exists a $2n$-component tangle T with skeleton as below such that $\tau(T) = K$ and where $\delta(T) = U$ is the *untangle*:



Hear more at ωεβ/AKT.

Acknowledgement. This work was supported by NSERC grant RGPIN-2018-04350 and by the Chu Family Foundation (NYC).

Jones: Formulas stay; interpretations change with time.



Formulas. Draw an n-crossing knot K as on the right: all crossings face up, and the edges are marked with a running index $k \in \{1, \dots, 2n+1\}$ and with rotation numbers φ_k. Let A be the $(2n+1)\times(2n+1)$ matrix constructed by starting with the identity matrix I, and adding a 2×2 block for each crossing:



A	col $i+1$	col $j+1$
row i	$-T^s$	$T^s - 1$
row j	0	-1

Let $G = (g_{\alpha\beta}) = A^{-1}$. For the trefoil example, it is:

$$A = \begin{pmatrix} 1 & -T & 0 & 0 & T-1 & 0 & 0 \\ 0 & 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & -T & 0 & 0 & T-1 \\ 0 & 0 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & T-1 & 0 & 1 & -T & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & -1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix},$$

$$G = \begin{pmatrix} 1 & T & 1 & T & 1 & T & 1 \\ 0 & 1 & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T^2}{T^2-T+1} & 1 \\ 0 & 0 & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T^2}{T^2-T+1} & 1 \\ 0 & 0 & \frac{1-T}{T^2-T+1} & \frac{1}{T^2-T+1} & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & 1 \\ 0 & 0 & \frac{1-T}{T^2-T+1} & -\frac{(T-1)T}{T^2-T+1} & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & 1 \\ 0 & 0 & 0 & 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

"The Green Function"

Burau Alexander Fox Wirtinger Blanchfield

Note. The Alexander polynomial Δ is given by

$$\Delta = T^{(-\varphi-w)/2}\det(A), \qquad \text{with } \varphi = \sum_k \varphi_k,\ w = \sum_c s.$$

Classical Topologists: This is boring. Yawn.

Formulas, continued. Finally, set

$$R_1(c) := s\left(g_{ji}\left(g_{j+1,j} + g_{j,j+1} - g_{ij}\right) - g_{ii}\left(g_{j,j+1} - 1\right) - 1/2\right)$$

$$\rho_1 := \Delta^2\left(\sum_c R_1(c) - \sum_k \varphi_k\left(g_{kk} - 1/2\right)\right).$$

In our example $\rho_1 = -T^2 + 2T - 2 + 2T^{-1} - T^{-2}$.

Theorem. ρ_1 is a knot invariant. Proof: later.

Classical Topologists: Whiskey Tango Foxtrot?

Cars, Interchanges, and Traffic Counters. Cars always drive forward. When a car crosses over a bridge it goes through with (algebraic) probability $T^s \sim 1$, but falls off with probability $1 - T^s \sim 0^*$. At the very end, cars fall off and disappear. See also [Jo, LTW].

Jones Lin Tian Wang

$p = 1 - T^s$



image credits: diamondtraffic.com

image credits: Dall-E

* In algebra $x \sim 0$ if for every y in the ideal generated by x, $1 - y$ is invertible.

References.

[BV1] D. Bar-Natan and R. van der Veen, *A Polynomial Time Knot Polynomial,* Proc. Amer. Math. Soc. **147** (2019) 377–397, arXiv:1708.04853.

[BV2] D. Bar-Natan and R. van der Veen, *Perturbed Gaussian Generating Functions for Universal Knot Invariants,* arXiv:2109.02057.

[Dr] V. G. Drinfel'd, *Quantum Groups,* Proc. Int. Cong. Math., 798–820, Berkeley, 1986.

[Jo] V. F. R. Jones, *Hecke Algebra Representations of Braid Groups and Link Polynomials,* Annals Math., **126** (1987) 335-388.

[La] R. J. Lawrence, *Universal Link Invariants using Quantum Groups,* Proc. XVII Int. Conf. on Diff. Geom. Methods in Theor. Phys., Chester, England, August 1988. World Scientific (1989) 55–63.

[LTW] X-S. Lin, F. Tian, and Z. Wang, *Burau Representation and Random Walk on String Links,* Pac. J. Math., **182-2** (1998) 289–302, arXiv:q-alg/9605023.

[Oh1] T. Ohtsuki, *Quantum Invariants,* Series on Knots and Everything **29**, World Scientific 2002.

[Oh2] T. Ohtsuki, *On the 2–loop Polynomial of Knots,* Geom. Top. **11** (2007) 1357–1475.

[Ov] A. Overbay, *Perturbative Expansion of the Colored Jones Polynomial,* Ph.D. thesis, University of North Carolina, August 2013, ωεβ/Ov.

[Ro1] L. Rozansky, *A Contribution of the Trivial Flat Connection to the Jones Polynomial and Witten's Invariant of 3D Manifolds, I,* Comm. Math. Phys. **175-2** (1996) 275–296, arXiv:hep-th/9401061.

[Ro2] L. Rozansky, *The Universal R-Matrix, Burau Representation and the Melvin-Morton Expansion of the Colored Jones Polynomial,* Adv. Math. **134-1** (1998) 1–31, arXiv:q-alg/9604005.

[Ro3] L. Rozansky, *A Universal U(1)-RCC Invariant of Links and Rationality Conjecture,* arXiv:math/0201139.

[Sch] S. Schaveling, *Expansions of Quantum Group Invariants,* Ph.D. thesis, Universiteit Leiden, September 2020, ωεβ/Scha.

Preliminaries

This is Rho.nb of http://drorbn.net/oa22/ap.

```
Once[<< KnotTheory`; << Rot.m];
```

```
Loading KnotTheory` version
  of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
Loading Rot.m from http://drorbn.net/la22/ap
  to compute rotation numbers.
```

The Program

$$R_1[s\_, i\_, j\_] := s\,(g_{ji}\,(g_{j^+,j} + g_{j,j^+} - g_{ij}) - g_{ii}\,(g_{j,j^+} - 1) - 1/2);$$

$$\begin{aligned}
&Z[K_] := \text{Module}\Big[\{Cs, \varphi, n, A, s, i, j, k, \Delta, G, \rho1\},\\
&\quad \{Cs, \varphi\} = \text{Rot}[K];\ n = \text{Length}[Cs];\\
&\quad A = \text{IdentityMatrix}[2n+1];\\
&\quad \text{Cases}\Big[Cs, \{s_, i_, j_\} \mapsto \\
&\qquad \Big(A[\![\{i, j\}, \{i+1, j+1\}]\!] \mathrel{+}= \begin{pmatrix} -T^s & T^s - 1 \\ 0 & -1 \end{pmatrix}\Big)\Big];\\
&\quad \Delta = T^{(-\text{Total}[\varphi] - \text{Total}[Cs[\![All,1]\!]])/2}\, \text{Det}[A];\\
&\quad G = \text{Inverse}[A];\\
&\quad \rho1 = \sum_{k=1}^{n} R_1 @@ Cs[\![k]\!] - \sum_{k=1}^{2n} \varphi[\![k]\!]\,(g_{kk} - 1/2);\\
&\quad \text{Factor@}\\
&\qquad \{\Delta, \Delta^2\rho1\ /.\ \alpha_^+ \mapsto \alpha + 1\ /.\ g_{\alpha_,\beta_} \mapsto G[\![\alpha, \beta]\!]\}\Big];
\end{aligned}$$

The First Few Knots

$$\text{TableForm}\Big[\text{Table}\big[\text{Join}\big[\{K[\![1]\!]_{K[\![2]\!]}\}, Z[K]\big], \{K, \text{AllKnots}[\{3, 6\}]\}\big], \text{TableAlignments} \to \text{Center}\Big]$$

3_1	$\frac{1-T+T^2}{T}$	$\frac{(-1+T)^2\,(1+T^2)}{T^2}$
4_1	$-\frac{1-3T+T^2}{T}$	0
5_1	$\frac{1-T+T^2-T^3+T^4}{T^2}$	$\frac{(-1+T)^2\,(1+T^2)\,(2+T^2+2T^4)}{T^4}$
5_2	$\frac{2-3T+2T^2}{T}$	$\frac{(-1+T)^2\,(5-4T+5T^2)}{T^2}$
6_1	$-\frac{(-2+T)\,(-1+2T)}{T}$	$\frac{(-1+T)^2\,(1-4T+T^2)}{T^2}$
6_2	$-\frac{1-3T+3T^2-3T^3+T^4}{T^2}$	$\frac{(-1+T)^2\,(1-4T+4T^2-4T^3+4T^4-4T^5+T^6)}{T^4}$
6_3	$\frac{1-3T+5T^2-3T^3+T^4}{T^2}$	0

$p = 1 - T^s$

Fast!



$$\begin{aligned}
&\text{Timing@}\\
&\ Z\big[GST48 = \text{EPD}\big[X_{14,1}, \bar{X}_{2,29}, X_{3,40}, X_{43,4}, \bar{X}_{26,5}, X_{6,95},\\
&\quad X_{96,7}, X_{13,8}, \bar{X}_{9,28}, X_{10,41}, X_{42,11}, \bar{X}_{27,12}, X_{30,15},\\
&\quad \bar{X}_{16,61}, \bar{X}_{17,72}, \bar{X}_{18,83}, X_{19,34}, \bar{X}_{89,20}, \bar{X}_{21,92},\\
&\quad \bar{X}_{79,22}, \bar{X}_{68,23}, \bar{X}_{57,24}, \bar{X}_{25,56}, X_{62,31}, X_{73,32},\\
&\quad X_{84,33}, \bar{X}_{50,35}, X_{36,81}, X_{37,70}, X_{38,59}, \bar{X}_{39,54}, X_{44,55},\\
&\quad X_{58,45}, X_{69,46}, X_{80,47}, X_{48,91}, X_{90,49}, X_{51,82}, X_{52,71},\\
&\quad X_{53,60}, \bar{X}_{63,74}, \bar{X}_{64,85}, \bar{X}_{76,65}, \bar{X}_{87,66}, \bar{X}_{67,94},\\
&\quad \bar{X}_{75,86}, \bar{X}_{88,77}, \bar{X}_{78,93}\big]\big]
\end{aligned}$$

$$\begin{aligned}
&\Big\{170.313, \Big\{-\frac{1}{T^8}\left(-1+2T-T^2-T^3+2T^4-T^5+T^8\right)\\
&\quad \left(-1+T^3-2T^4+T^5+T^6-2T^7+T^8\right), \frac{1}{T^{16}}\\
&\quad (-1+T)^2\,\big(5-18T+33T^2-32T^3+2T^4+42T^5-62T^6-\\
&\qquad 8T^7+166T^8-242T^9+108T^{10}+132T^{11}-226T^{12}+\\
&\qquad 148T^{13}-11T^{14}-36T^{15}-11T^{16}+148T^{17}-226T^{18}+\\
&\qquad 132T^{19}+108T^{20}-242T^{21}+166T^{22}-8T^{23}-62T^{24}+\\
&\qquad 42T^{25}+2T^{26}-32T^{27}+33T^{28}-18T^{29}+5T^{30}\big)\Big\}\Big\}
\end{aligned}$$

Strong!

```
{NumberOfKnots[{3, 12}],
 Length@
  Union@Table[Z[K], {K, AllKnots[{3, 12}]}],
 Length@
  Union@Table[{HOMFLYPT[K], Kh[K]},
    {K, AllKnots[{3, 12}]}]}
```

```
{2977, 2882, 2785}
```

So the pair (Δ, ρ_1) attains 2,882 distinct values on the 2,977 prime knots with up to 12 crossings (a deficit of 95), whereas the pair (HOMFLYPT, Khovanov Homology) attains only 2,785 distinct values on the same knots (a deficit of 192).

Hoste Ocneanu Millett Freyd Lickorish Yetter Przytycki Traczyk Khovanov

Theorem. The Green function $g_{\alpha\beta}$ is the reading of a traffic counter at β, if car traffic is injected at α (if $\alpha=\beta$, the counter is *after* the injection point).



Example.

$\sum_{p\geq 0}(1-T)^p = T^{-1}$



$$G=\begin{pmatrix}1 & T^{-1} & 1\\ 0 & T^{-1} & 1\\ 0 & 0 & 1\end{pmatrix}$$

Proof. Near a crossing c with sign s, incoming upper edge i and incoming lower edge j, both sides satisfy the g-*rules*:



$$g_{i\beta}=\delta_{i\beta}+T^s g_{i+1,\beta}+(1-T^s)g_{j+1,\beta},\qquad g_{j\beta}=\delta_{j\beta}+g_{j+1,\beta},$$

and always, $g_{\alpha,2n+1}=1$: use common sense and $AG=I\ (=GA)$.

Bonus. Near c, both sides satisfy the further g-*rules*:

$$g_{\alpha i}=T^{-s}(g_{\alpha,i+1}-\delta_{\alpha,i+1}),\qquad g_{\alpha j}=g_{\alpha,j+1}-(1-T^s)g_{\alpha i}-\delta_{\alpha,j+1}.$$

Invariance of ρ_1. We start with the hardest, Reidemeister 3:



⇒ Overall traffic patterns are unaffected by Reid3!

⇒ Green's $g_{\alpha\beta}$ is unchanged by Reid3, provided the cars injection site α and the traffic counters β are away.

⇒ Only the contribution from the R_1 terms within the Reid3 move matters, and using g-rules the relevant $g_{\alpha\beta}$'s can be pushed outside of the Reid3 area:




```
δi_,j_ := If[i === j, 1, 0];
gRuless_,i_,j_ :=
 {giβ_ :→ δiβ + T^s gi+,β + (1 - T^s) gj+,β, gjβ_ :→ δjβ + gj+,β,
  gα_,i :→ T^-s (gα,i+ - δα,i+),
  gα_j :→ gα,j+ - (1 - T^s) gαi - δα,j+}
lhs = R1[1, j, k] + R1[1, i, k+] + R1[1, i+, j+] //.
   gRules1,j,k ⋃ gRules1,i,k+ ⋃ gRules1,i+,j+;
rhs = R1[1, i, j] + R1[1, i+, k] + R1[1, j+, k+] //.
   gRules1,i,j ⋃ gRules1,i+,k ⋃ gRules1,j+,k+;
Simplify[lhs == rhs]
True
```

Next comes Reid1, where we use results from an earlier example:

```
R1[1, 2, 1] - 1 (g22 - 1/2) /. gα_,β_ :→ (1 T^-1 1
                                           0 T^-1 1
                                           0  0   1)[[α, β]]
```

$$\frac{1}{T^2}-\frac{1}{T}-\frac{-1+\frac{1}{T}}{T}\ = 0$$



Invariance under the other moves is proven similarly.

Wearing my Topology hat the formula for R_1, and even the idea to look for R_1, remain a complete mystery to me.

Wearing my Quantum Algebra hat, I spy a Heisenberg algebra $\mathbb{H}=A\langle p,x\rangle/([p,x]=1)$:

cars ↔ p traffic counters ↔ x

Where did it come from? Consider $\mathfrak{g}_\epsilon := sl_{2+}^\epsilon := L\langle y,b,a,x\rangle$ with relations

$$[b,x]=\epsilon x,\quad [b,y]=-\epsilon y,\quad [b,a]=0,$$
$$[a,x]=x,\quad [a,y]=-y,\quad [x,y]=b+\epsilon a.$$

At invertible ϵ, it is isomorphic to sl_2 plus a central factor, and it can be quantized à la Drinfel'd [Dr] much like sl_2 to get an algebra $QU=A\langle y,b,a,x\rangle$ subject to (with $q=e^{\hbar\epsilon}$):

$$[b,a]=0,\quad [b,x]=\epsilon x,\quad [b,y]=-\epsilon y,$$
$$[a,x]=x,\quad [a,y]=-y,\quad xy-qyx=\frac{1-e^{-\hbar(b+\epsilon a)}}{\hbar}.$$

Now QU has an R-matrix solving Yang-Baxter (meaning Reid3),

$$R=\sum_{m,n\geq 0}\frac{y^n b^m\otimes(\hbar a)^m(\hbar x)^n}{m![n]_q!},\qquad ([n]_q!\text{ is a "quantum factorial"})$$

and so it has an associated "universal quantum invariant" à la Lawrence and Ohtsuki [La, Oh1], $Z_\epsilon(K)\in QU$.

Now $QU\cong\mathcal{U}(\mathfrak{g}_\epsilon)$ (only as algebras!) and $\mathcal{U}(\mathfrak{g}_\epsilon)$ represents into $\mathbb{H}$ via

$$y\to -tp-\epsilon\cdot xp^2,\quad b\to t+\epsilon\cdot xp,\quad a\to xp,\quad x\to x,$$

(abstractly, $\mathfrak{g}_\epsilon$ acts on its Verma module

$$\mathcal{U}(\mathfrak{g}_\epsilon)/(\mathcal{U}(\mathfrak{g}_\epsilon)\langle y,a,b-\epsilon a-t\rangle)\cong\mathbb{Q}[x]$$

by differential operators, namely via $\mathbb{H}$), so R can be pushed to $\mathcal{R}\in\mathbb{H}\otimes\mathbb{H}$.

Everything still makes sense at $\epsilon=0$ and can be expanded near $\epsilon=0$ resulting with $\mathcal{R}=\mathcal{R}_0(1+\epsilon\mathcal{R}_1+\cdots)$, with $\mathcal{R}_0=e^{t(xp\otimes 1-x\otimes p)}$ and $\mathcal{R}_1$ a quartic polynomial in p and x. So p's and x's get created along K and need to be pushed around to a standard location ("normal ordering"). This is done using

$$(p\otimes 1)\mathcal{R}_0=\mathcal{R}_0(T(p\otimes 1)+(1-T)(1\otimes p)),$$
$$(1\otimes p)\mathcal{R}_0=\mathcal{R}_0(1\otimes p),$$

and when the dust settles, we get our formulas for ρ_1. But QU is a quasi-triangular Hopf algebra, and hence **ρ_1 is homomorphic**. Read more at [BV1, BV2] and hear more at ωεβ/SolvApp, ωεβ/Dogma, ωεβ/DoPeGDO, ωεβ/FDA, ωεβ/AQDW.


Schaveling

Also, we can (and know how to) look at higher powers of ϵ and we can (and more or less know how to) replace sl_2 by arbitrary semi-simple Lie algebra (e.g., [Sch]). So **ρ_1 is not alone!**

These constructions are very similar to Rozansky-Overbay [Ro1, Ro2, Ro3, Ov] and hence to the "loop expansion" of the Kontsevich integral and the coloured Jones polynomial [Oh2].

If this all reads like **insanity** to you, it should (and you haven't seen half of it). Simple things should have simple explanations. Hence, **Homework.** Explain ρ_1 with no reference to quantum voodoo and find it a topology home (large enough to house generalizations!). Make explicit the homomorphic properties of ρ_1. Use them to do topology!

P.S. As a friend of Δ, ρ_1 gives a genus bound, sometimes better than Δ's. How much further does this friendship extend?

A Small-Print Page on ρ_d, $d > 1$.

Definition. $\langle f(z_i), h(\zeta_i)\rangle_{\{z_i\}} := f(\partial_{\zeta_i})h\big|_{\zeta_i=0}$, so $\langle p^2x^2, e^{g\pi\xi}\rangle = 2g^2$.

Baby Theorem. There exist (non unique) power series $r^\pm(p_1,p_2,x_1,x_2) = \sum_d \epsilon^d r^\pm_d(p_1,p_2,x_1,x_2) \in \mathbb{Q}[T^{\pm1},p_1,p_2,x_1,x_2][\![\epsilon]\!]$ with $\deg r^\pm_d \leq 2d+2$ ("docile") such that the power series $Z^b = \sum \rho^b_d \epsilon^d :=$

$$\left\langle \exp\left(\sum_c r^s(p_i,p_j,x_i,x_j)\right), \exp\left(\sum_{\alpha,\beta} g_{\alpha\beta}\pi_\alpha\xi_\beta\right)\right\rangle_{\{p_\alpha,x_\beta\}}$$

is a bnot invariant. Beyond the once-and-for-all computation of $g_{\alpha\beta}$ (a matrix inversion), Z^b is computable in $O(n^d)$ operations in the ring $\mathbb{Q}[T^{\pm1}]$.

(Bnots are knot diagrams modulo the braid-like Reidemeister moves, but not the cyclic ones).

Theorem. There also exist docile power series $\gamma^\varphi(\bar p,\bar x) = \sum_d \epsilon^d\gamma^\varphi_d \in \mathbb{Q}[T^{\pm1},\bar p,\bar x][\![\epsilon]\!]$ such that the power series $Z = \sum \rho_d\epsilon^d :=$

$$\left\langle \exp\left(\sum_c r^s(p_i,p_j,x_i,x_j) + \sum_k \gamma^{\varphi_k}(\bar p_k,\bar x_k)\right),\right.$$
$$\left.\exp\left(\sum_{\alpha,\beta} g_{\alpha\beta}(\pi_\alpha+\bar\pi_\alpha)(\xi_\beta+\bar\xi_\beta) + \sum_\alpha \pi_\alpha\bar\xi_\alpha\right)\right\rangle_{\{p_\alpha,\bar p_\alpha,x_\beta,\bar x_\beta\}}$$

is a knot invariant, as easily computable as Z^b.

Implementation. Data, then program (with output using the Conway variable $z = \sqrt{T} - 1/\sqrt{T}$), and then a demo. See Rho.nb of ωεβ/ap.

```
V@γ_{1,φ_}[k_] = φ (1/2 - p̄_k x̄_k); V@γ_{2,φ_}[k_] = -φ^2 p̄_k x̄_k / 2;
V@γ_{3,φ_}[k_] := -φ^3 p̄_k x̄_k / 6
```

```
V@r_{1,s_}[i_, j_] :=
 s (-1 + 2 p_i x_i - 2 p_j x_i + (-1 + T^s) p_i p_j x_i^2 + (1 - T^s) p_j^2 x_i^2 - 2 p_i p_j x_i x_j + 2 p_j^2 x_i x_j) / 2
```

```
V@r_{2,1}[i_, j_] :=
 (-6 p_i x_i + 6 p_j x_i - 3 (-1 + 3 T) p_i p_j x_i^2 + 3 (-1 + 3 T) p_j^2 x_i^2 + 4 (-1 + T) p_i^2 p_j x_i^3 -
   2 (-1 + T) (5 + T) p_i p_j^2 x_i^3 + 2 (-1 + T) (3 + T) p_j^3 x_i^3 + 18 p_i p_j x_i x_j -
   18 p_j^2 x_i x_j - 6 p_i^2 p_j x_i^2 x_j + 6 (2 + T) p_i p_j^2 x_i^2 x_j - 6 (1 + T) p_j^3 x_i^2 x_j -
   6 p_i p_j^2 x_i x_j^2 + 6 p_j^3 x_i x_j^2) / 12
```

```
V@r_{2,-1}[i_, j_] :=
 (-6 T^2 p_i x_i + 6 T^2 p_j x_i + 3 (-3 + T) T p_i p_j x_i^2 - 3 (-3 + T) T p_j^2 x_i^2 -
   4 (-1 + T) T p_i^2 p_j x_i^3 + 2 (-1 + T) (1 + 5 T) p_i p_j^2 x_i^3 - 2 (-1 + T) (1 + 3 T) p_j^3 x_i^3 +
   18 T^2 p_i p_j x_i x_j - 18 T^2 p_j^2 x_i x_j - 6 T^2 p_i^2 p_j x_i^2 x_j + 6 T (1 + 2 T) p_i p_j^2 x_i^2 x_j -
   6 T (1 + T) p_j^3 x_i^2 x_j - 6 T^2 p_i p_j^2 x_i x_j^2 + 6 T^2 p_j^3 x_i x_j^2) / (12 T^2)
```

```
V@r_{3,1}[i_, j_] :=
 (4 p_i x_i - 4 p_j x_i + 2 (5 + 7 T) p_i p_j x_i^2 - 2 (5 + 7 T) p_j^2 x_i^2 - 4 (-5 + 6 T) p_i^2 p_j x_i^3 +
   4 (-16 + 17 T + 2 T^2) p_i p_j^2 x_i^3 - 4 (-11 + 11 T + 2 T^2) p_j^3 x_i^3 + 3 (-1 + T) p_i^3 p_j x_i^4 -
   3 (-1 + T) (4 + 3 T) p_i^2 p_j^2 x_i^4 + (-1 + T) (13 + 22 T + T^2) p_i p_j^3 x_i^4 -
   (-1 + T) (4 + 13 T + T^2) p_j^4 x_i^4 - 28 p_i p_j x_i x_j + 28 p_j^2 x_i x_j + 36 p_i^2 p_j x_i^2 x_j -
   12 (9 + 2 T) p_i p_j^2 x_i^2 x_j + 24 (3 + T) p_j^3 x_i^2 x_j - 4 p_i^3 p_j x_i^3 x_j + 28 T p_i^2 p_j^2 x_i^3 x_j -
   4 (-6 + 17 T + T^2) p_i p_j^3 x_i^3 x_j + 4 (-5 + 10 T + T^2) p_j^4 x_i^3 x_j + 24 p_i p_j^2 x_i x_j^2 -
   24 p_j^3 x_i x_j^2 - 24 p_i^2 p_j^2 x_i^2 x_j^2 + 6 (10 + T) p_i p_j^3 x_i^2 x_j^2 - 6 (6 + T) p_j^4 x_i^2 x_j^2 -
   4 p_i p_j^3 x_i x_j^3 + 4 p_j^4 x_i x_j^3) / 24
```

```
V@r_{3,-1}[i_, j_] :=
 (-4 T^3 p_i x_i + 4 T^3 p_j x_i - 2 T^2 (7 + 5 T) p_i p_j x_i^2 + 2 T^2 (7 + 5 T) p_j^2 x_i^2 -
   4 T^2 (-6 + 5 T) p_i^2 p_j x_i^3 + 4 T (-2 - 17 T + 16 T^2) p_i p_j^2 x_i^3 -
   4 T (-2 - 11 T + 11 T^2) p_j^3 x_i^3 + 3 (-1 + T) T^2 p_i^3 p_j x_i^4 - 3 (-1 + T) T (3 + 4 T) p_i^2 p_j^2 x_i^4 +
   (-1 + T) (1 + 22 T + 13 T^2) p_i p_j^3 x_i^4 - (-1 + T) (1 + 13 T + 4 T^2) p_j^4 x_i^4 +
   28 T^3 p_i p_j x_i x_j - 28 T^3 p_j^2 x_i x_j - 36 T^3 p_i^2 p_j x_i^2 x_j + 12 T^2 (2 + 9 T) p_i p_j^2 x_i^2 x_j -
   24 T^2 (1 + 3 T) p_j^3 x_i^2 x_j + 4 T^3 p_i^3 p_j x_i^3 x_j - 28 T^2 p_i^2 p_j^2 x_i^3 x_j -
   4 T (-1 - 17 T + 6 T^2) p_i p_j^3 x_i^3 x_j + 4 T (-1 - 10 T + 5 T^2) p_j^4 x_i^3 x_j -
   24 T^3 p_i p_j^2 x_i x_j^2 + 24 T^3 p_j^3 x_i x_j^2 + 24 T^3 p_i^2 p_j^2 x_i^2 x_j^2 - 6 T^2 (1 + 10 T) p_i p_j^3 x_i^2 x_j^2 +
   6 T^2 (1 + 6 T) p_j^4 x_i^2 x_j^2 + 4 T^3 p_i p_j^3 x_i x_j^3 - 4 T^3 p_j^4 x_i x_j^3) / (24 T^3)
```

```
{p*, x*, p̄*, x̄*} = {π, ξ, π̄, ξ̄}; (z_i__)* := (z*)_i;
Zip_{}[ℰ_] := ℰ;
Zip_{z_,zs___}[ℰ_] :=
 (Collect[ℰ // Zip_{zs}, z] /. f_. z^d_. :→ (D[f, {z*, d}])) /. z* → 0
```

```
gPair[fs_, w_] :=
 gPair[fs, w] =
  Collect[Zip_{Join@@Table[{p_α, p̄_α, x_α, x̄_α}, {α, w}]}[
    (Times @@ (V /@ fs))
     Exp[Sum[g_{α,β} (π_α + π̄_α) (ξ_β + ξ̄_β), {α, w}, {β, w}] - Sum[ξ̄_α π_α, {α, w}]]],
   g__, Factor]
```

```
T2z[p_] := Module[{q = Expand[p], n, c},
  If[q === 0, 0, c = Coefficient[q, T, n = Exponent[q, T]];
   c z^(2 n) + T2z[q - c (T^(1/2) - T^(-1/2))^(2 n)]]];
```

```
Z_d_[K_] := Module[{Cs, φ, n, A, s, i, j, k, Δ, G, d1, Z1, Z2, Z3},
  {Cs, φ} = Rot[K]; n = Length[Cs]; A = IdentityMatrix[2 n + 1];
  Cases[Cs, {s_, i_, j_} :→ (A⟦{i, j}, {i + 1, j + 1}⟧ += ({{-T^s, T^s - 1}, {0, -1}}))];
  {Δ, G} = Factor@{T^((-Total[φ] - Total[Cs⟦All,1⟧])/2) Det@A, Inverse@A};
  Z1 =
   Exp[Total[Cases[Cs, {s_, i_, j_} :→ Sum[ϵ^d1 r_{d1,s}[i, j], {d1, d}]]] +
     Sum[ϵ^d1 γ_{d1,φ⟦k⟧}[k], {k, 2 n}, {d1, d}] /. γ_{_,0}[_] → 0];
  Z2 = Expand[F[{}, {}] × Normal@Series[Z1, {ϵ, 0, d}]] //.
    F[fs_, {es___}] × (f : (r | γ)_ps__[is__])^p_. :→
     F[Join[fs, Table[f, p]], DeleteDuplicates@{es, is}];
  Z3 = Expand[Z2 /. F[fs_, es_] :→ Expand[gPair[
        Replace[fs, Thread[es → Range@Length@es], {2}], Length@es
       ] /. g_{α_,β_} :→ G⟦es⟦α⟧, es⟦β⟧⟧]];
  Collect[{Δ, Z3 /. ϵ^p_. → p! Δ^(2 p) ϵ^p}, ϵ, T2z]];
```

```
Z_2[GST48] (* takes a few minutes *)
```

{1 − 4 z^2 − 61 z^4 − 207 z^6 − 296 z^8 − 210 z^10 − 77 z^12 − 14 z^14 − z^16,
 1 + (38 z^2 + 255 z^4 + 1696 z^6 + 16 281 z^8 + 86 952 z^10 + 259 994 z^12 + 487 372 z^14 + 615 066 z^16 + 543 148 z^18 + 341 714 z^20 +
 153 722 z^22 + 48 983 z^24 + 10 776 z^26 + 1554 z^28 + 132 z^30 + 5 z^32) ϵ +
 (−8 − 484 z^2 + 9709 z^4 + 165 952 z^6 + 1 590 491 z^8 + 16 256 508 z^10 + 115 341 797 z^12 + 432 685 748 z^14 + 395 838 354 z^16 − 4 017 557 792 z^18 − 23 300 064 167 z^20 −
 70 082 264 972 z^22 − 142 572 271 191 z^24 − 209 475 503 700 z^26 − 221 616 295 209 z^28 − 151 502 648 428 z^30 − 23 700 199 243 z^32 +
 99 462 146 328 z^34 + 164 920 463 074 z^36 + 162 550 825 432 z^38 + 119 164 552 296 z^40 + 69 153 062 608 z^42 + 32 547 596 611 z^44 + 12 541 195 448 z^46 +
 3 961 384 155 z^48 + 1 021 219 696 z^50 + 212 773 106 z^52 + 35 264 208 z^54 + 4 537 548 z^56 + 436 600 z^58 + 29 536 z^60 + 1252 z^62 + 25 z^64) ϵ^2}

```
TableForm[Table[Join[{K⟦1⟧_{K⟦2⟧}}, Z_3[K]], {K, AllKnots[{3, 6}]}], TableAlignments → Center] (* takes a few minutes *)
```

3_1	$1+z^2$	$1+(2z^2+z^4)\epsilon+(2-4z^2+3z^4+4z^6+z^8)\epsilon^2+(-12+74z^2-27z^4-20z^6+8z^8+6z^{10}+z^{12})\epsilon^3$
4_1	$1-z^2$	$1+(-2+2z^4)\epsilon^2$
5_1	$1+3z^2+z^4$	$1+(10z^2+21z^4+12z^6+2z^8)\epsilon+(6-28z^2+33z^4+364z^6+655z^8+536z^{10}+227z^{12}+48z^{14}+4z^{16})\epsilon^2+(-60+970z^2+645z^4-3380z^6-3280z^8+7470z^{10}+19475z^{12}+20536z^{14}+12564z^{16}+4774z^{18}+1109z^{20}+144z^{22}+8z^{24})\epsilon^3$
5_2	$1+2z^2$	$1+(6z^2+5z^4)\epsilon+(4-20z^2+43z^4+64z^6+26z^8)\epsilon^2+(-36+498z^2-883z^4+100z^6+816z^8+556z^{10}+146z^{12})\epsilon^3$
6_1	$1-2z^2$	$1+(-2z^2+z^4)\epsilon+(-4+4z^2+25z^4-8z^6+2z^8)\epsilon^2+(12+154z^2-223z^4-608z^6+100z^8-52z^{10}+10z^{12})\epsilon^3$
6_2	$1-z^2-z^4$	$1+(-2z^2-3z^4+2z^6+z^8)\epsilon+(-2-4z^2+29z^4+28z^6+42z^8-8z^{10}-2z^{12}+4z^{14}+z^{16})\epsilon^2+(12+166z^2+155z^4-194z^6-2453z^8-1622z^{10}-1967z^{12}-258z^{14}+49z^{16}-30z^{18}+z^{20}+6z^{22}+z^{24})\epsilon^3$
6_3	$1+z^2+z^4$	$1+(2+8z^2-16z^6-24z^8-16z^{10}-2z^{12})\epsilon^2$